CIS 425 - Review of Datatypes and Union types

To define a datatype in ML, we need to give the name and the constructor of the datatype.

```
datatype 'a myList = empty | B of 'a * 'a myList;
```

As we can see in this example, we give the name of the datatype `myList` and the constructors are `empty` and `B` :

```
empty;
val it = empty : 'a myList
B;
val it = fn : 'a * 'a myList -> 'a myList
```

As we can see here, we can give our datatype a polymorphic type: this list datatype that we just constructed could be a list of ints, reals, bools, or even any other datatype that we create.

We can make a list of ints

```
val a = B(5,empty);
```

or a list of bools.

```
val a = B(true, empty);
```

However, we cannot have a list of multiple datatypes. Once we have set α to a certain datatype, it must stay consistent.

```
val a = B(5, B(true, empty)); (* This will give an error! *)
```

However, we might want to be able to make a list of multiple datatypes. How would we be able to make a list of ints and bools?

To do that, we will need a new datatype! Just as how we invented the category "a Piece of Chex Mix" to include cereal, pretzels, bagel chips, and whatever else they put in chex mix, we can invent our own Bool and Int Mix.

```
datatype mix = I of int | C of bool;
```

(We are, slightly awkwardly, using `C` instead of `B`, since we already used `B` in our definition of the `myList` datatype.)

Now that we have our mix defined, we can finally represent our list of bools and ints.

```
val a = B(I 5, B(C true, empty)); (* No errors this time! *)
```

1 Map

We want to define the map function: that is a function that takes an inner function and a list, and returns a list where each element has that inner function applied to it.

In ML, we define functions using pattern matching. Recall our myList datatype. There were two patterns that defined that datatype: the empty list, and the list with an initial element paired with the rest of the list. For map, there are two patterns that we have to match: the empty list, and the list with a first element joined to the remainder of the list.

```
fun map f [] = (* something *)
  | map f (x::xs) = (* something else *)
```

With map, we want to apply `f` to every element of the list. If we apply `f` to the empty list `[]`, we should get back the empty list, since there are no elements. On the other hand, we can apply `f` to a non-empty list recursively, by applying it to the first element, and then calling map on the remainder of the list. Our implementation looks like this.

```
fun map f [] = []
  | map f (x::xs) = (f x) :: map f xs;
```

We could just as easily have passed the arguments as a tuple, in which case our function would look like this.

```
fun map (f, []) = []
  | map (f, (x::xs)) = (f x) :: map(f, xs);
```

The first function is curried, the second is uncurried. Note that both functions take just one argument. The curried version of `map` returns a function that acts on the list, as shown below:

```
fun map f   =   fn []         => []
                 |   x :: xs    => (f x) :: map f xs;
```

while the uncurried version takes arguments as a single tuple.

1.1 Digression 1: Syntactic Sugar

When we talk about the list `[1,2,3]` in ML, this is really syntactic sugar for `1::2::3::nil`. Syntactic sugar is notation that makes programs easier to read and write, but doesn't actually add any functionality to the programming language.

1.2 Digression 2: Advantages of Partial Computations

Suppose we have the following function and function call.

```
fun f x y = x + 1;
val a = f 3;
```

What is the value of a? It is bound to the function `fn y => 3+1`. However, at the point the compiler could apply some optimizations and execute the addition instead of generating code for it. In this way, `a` will be bound to the function `fn y => 4`. That is, the computation inside of function f that did not depend on y has been computed, before we get the value of y. This way, we can call `a` as many times as we want, and it will not require performing the calculation many times.

2 Reduce

Like the map function, the reduce function will apply another function to every element of a list. However, this time, for each element of the list we want to update the value of another variable with the result of the function applied to the element, and that variable. Like the map function, we will need to consider the base case of the empty list and the inductive case of the non-empty list. Here is a proposed version of the reduce function:

```
fun reduce f [] a = a
  | reduce f (x::xs) a = f(x,a)::reduce f xs a; (* Will give us an
    error! *)
```

This function is wrong! How can we tell? One easy way is by considering the types. The base case will give us an int, the type of `a`, whereas the inductive case will give us a list (which isn't what we want reduce to give us anyway). In ML, the types of the different cases must match. The correct function is shown below.

```
fun reduce f [] a = a
  | reduce f (x::xs) a = reduce f xs (f(x,a));
```

(There are two ways of defining reduce: reducing from the left end of the list, and reducing from the right end of the list. This is one of them, you could try to find the other if you want a "`fun`" exercise.

3 What's the difference between a datatype and a C union type?

Let's consider a C union type, which we will ominously call `unsafe`.

```
union {
    float f;
    int i;
} unsafe;

unsafe.f = 1.0;
printf("%d\n", unsafe.i + 2);
```

What does this print? Did you guess 1065353218? What we just did was read a float as an int, and the compiler raised no errors.

This kind of behavior is not allowed in the high-level, lawful palace known as ML. Consider our Chex Mix datatype from above, where we had both ints and bools. The datatype is not quite the union of the int and bool domain, but is what the set theorists call a "disjoint union," notated as int⊎bool instead of int∪bool. This means that if we take an element from the datatype, we will be able to know if it's an int or a bool: The constructors in an ML datatype are represented at a lower level as a bit or bits called "tags," set aside to tell us which type it is. The same program written in ML will raise an exception:

```
datatype safe = I of int | R of real; (* The SML response to the c
    union above *)
val safe = R 1.0;
print (Int.toString 5);
let val (I x) = safe
     in print (Int.toString (x+2))
end;

uncaught exception Bind
```